

Tema 4

Successions numèriques

Objectius

1. Definir successions amb wxMaxima.
2. Calcular elements d'una successió.
3. Realitzar operacions amb successions.
4. Interpretar la definició de límit d'una successió.
5. Calcular el límit d'una successió.
6. Aplicació pràctica dels criteris de convergència i de càlcul de límits.
7. Aplicació pràctica de les successions indeterminades o indeterminacions.
8. Aplicació pràctica del concepte de successions equivalents i les seves aplicacions al càlcul de límits.

Continguts

- 04-1. Definició de successions. Operacions amb successions.
- 04-2. Càlcul del límit d'una successió.
- 04-3. Formes indeterminades o indeterminacions
- 04-4. Criteris de convergència i càlcul de límits de successions numèriques.
- 04-5. Equivalència de successions.

Referències

- AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.
- APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

- BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima
- BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).
- ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.
- ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable
- JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.
- RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima
- RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería
- SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).
- VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

04-1.- Definició de successions. Operacions amb successions

Els càlculs corresponents a aquest apartat es desenvolupen al fitxer **Tema_04-1.wxm**.

La manera habitual d'expressar una successió és mitjançant una funció de variable discreta, és a dir, una funció definida com una aplicació del conjunt dels nombres naturals en un subconjunt dels nombres reals, tal que a cada nombre natural $n \in \mathbb{N}$ se li assigna un únic nombre real $x_n \in \mathbb{R}$. La notació habitual per a designar una successió és $(x_n)_{n \in \mathbb{N}}$ o simplement (x_n) . El nombre real x_n s'anomena terme general o terme n -èsim de la successió.

La sintaxi per a definir una successió amb wxMaxima és senzilla; cal assignar una denominació al terme general de la successió i escriure a continuació l'expressió d'aquest terme en funció del nombre natural n ; per exemple, la successió $\left(\frac{2n-1}{3n+2}\right)_{n \in \mathbb{N}}$ es defineix mitjançant:

```
(%i1) s01[n]:=(2*n-1)/(3*n+2);  
(%o1) s01_n :=  $\frac{2n-1}{3n+2}$ 
```

Es poden obtenir alguns termes de la successió, explicitant el nombre natural antiimatge del terme de la successió; per exemple, els cinc primers termes de la successió anterior són:

```
(%i2) s01[1];    s01[10];    s01[20];  
(%o2)  $\frac{1}{5}$   
(%o3)  $\frac{19}{32}$   
(%o4)  $\frac{39}{62}$ 
```

Si es vol obtenir una llista amb uns quants termes de la successió, això es pot obtenir mitjançant la instrucció “makelist”, que es pot concretar amb la instrucció “numer” per indicar que es volen els termes en representació decimal; així, per exemple:

```
(%i5) makelist(s01[n], n, 1, 10);  
(%o5) [ $\frac{1}{5}, \frac{3}{8}, \frac{5}{11}, \frac{1}{2}, \frac{9}{17}, \frac{11}{20}, \frac{13}{23}, \frac{15}{26}, \frac{17}{29}, \frac{19}{32}$ ]  
(%i6) makelist(s01[n], n, 1, 5),numer;  
(%o6) [0.2, 0.375, 0.45454545454545, 0.5, 0.52941176470588]
```

Altres exemples de successions i la seva definició amb wxMaxima:

```
(%i7) s02[n]:=sin(1/n);
(%o7) s02_n:=sin( $\frac{1}{n}$ )

(%i8) makelist(s02[n], n, 1,5);
(%o8) [sin(1), sin( $\frac{1}{2}$ ), sin( $\frac{1}{3}$ ), sin( $\frac{1}{4}$ ), sin( $\frac{1}{5}$ )]

(%i9) s03[n]:=sin(n);
(%o9) s03_n:=sin(n)

(%i10) makelist(s03[n], n, 1, 5);
(%o10) [sin(1), sin(2), sin(3), sin(4), sin(5)]

(%i11) s04[n]:=(-1)^n;
(%o11) s04_n:=(-1)^n

(%i12) makelist(s04[n], n, 1, 5);
(%o12) [-1, 1, -1, 1, -1]

(%i13) s05[n]:=(-1)^n/n;
(%o13) s05_n:= $\frac{(-1)^n}{n}$ 

(%i14) makelist(s05[n], n, 1, 5);
(%o14) [-1,  $\frac{1}{2}$ ,  $-\frac{1}{3}$ ,  $\frac{1}{4}$ ,  $-\frac{1}{5}$ ]

(%i15) s06[n]:=n^2;
(%o15) s06_n:=n^2

(%i16) makelist(s06[n], n, 1, 5);
(%o16) [1, 4, 9, 16, 25]

(%i17) s07[n]:=-n^2+3;
(%o17) s07_n:=-n^2+3

(%i18) makelist(s07[n], n, 1, 5);
(%o18) [2, -1, -6, -13, -22]

(%i19) s08[n]:=1+(-1)^n/n;
(%o19) s08_n:=1+ $\frac{(-1)^n}{n}$ 

(%i20) makelist(s08[n], n, 1, 5);
(%o20) [0,  $\frac{3}{2}$ ,  $\frac{2}{3}$ ,  $\frac{5}{4}$ ,  $\frac{4}{5}$ ]
```

Una altra manera de definir o expressar una successió és per recurrència, que consisteix en donar una expressió del valor del terme general en funció d'un o més termes anteriors i concretar els valors inicials que calen per tenir correctament definida la successió. Una successió definida així es diu recurrent o definida per recurrència.

Per exemple, la successió definida per recurrència mitjançant:

$$\begin{cases} x_1 = 3 \\ x_n = \sqrt{2 + x_{n-1}}, \quad \forall n \geq 2 \end{cases}$$

es defineix amb wxMaxima de la forma següent:

```
(%i21) s10[1]:3$          s10[n]:=sqrt(2+s10[n-1]);
(%o22) s10_n:=sqrt(2+s10_{n-1})
(%i23) makelist(s10[n], n, 1, 5);
(%o23) [3, sqrt(5), sqrt(sqrt(5)+2), sqrt(sqrt(sqrt(5)+2)+2), sqrt(sqrt(sqrt(sqrt(5)+2)+2)+2)]
```

Un altre exemple de successió recurrent: la successió definida per:

$$\begin{cases} x_1 = 3 \\ x_n = \frac{1}{2} \left(x_{n-1} + \frac{3}{x_{n-1}} \right), \quad \forall n \geq 2 \end{cases}$$

En wxMaxima es defineix mitjançant:

```
(%i24) s12[1]:3$          s12[n]:=(1/2)*(s12[n-1]+3/s12[n-1]);
(%o25) s12_n:=1/2*(s12_{n-1}+3/s12_{n-1})
(%i26) makelist(s12[n], n, 1, 5);
(%o26) [3, 2, 7/4, 97/56, 18817/10864]
```

I ara un clàssic d'aquest tipus de successions: l'anomenada successió de Fibonacci, definida mitjançant:

$$\begin{cases} x_1 = 1, \quad x_2 = 1 \\ x_n = x_{n-2} + x_{n-1}, \quad \forall n \geq 3 \end{cases}$$

```
(%i27) FIB[1]:1;      FIB[2]:1;      FIB[n]:=FIB[n-1]+FIB[n-2];
(%o27) 1
(%o28) 1
(%o29) FIBn := FIBn-1 + FIBn-2
(%i30) makelist(FIB[n], n, 1, 10);
(%o30) [1, 1, 2, 3, 5, 8, 13, 21, 34, 55]
```

Es poden dur a terme operacions algebraiques amb successions, mitjançant operacions amb els seus termes. Aquestes operacions són:

Suma / diferència:	$(x_n) \pm (y_n) = (x_n \pm y_n)$
Producte:	$(x_n) \cdot (y_n) = (x_n y_n)$
Producte per escalar:	$\lambda \cdot (x_n) = (\lambda x_n), \lambda \in \mathbb{R}$
Quocient:	$(x_n) / (y_n) = (x_n / y_n), y_n \neq 0, \forall n \in \mathbb{N}$

La sintaxi per fer aquestes operacions amb wxMaxima és senzilla: es defineix una nova successió especificant les operacions i les successions que operen. A continuació es posen exemples de suma, diferència, combinació d'aquestes operacions, producte i quocient:

```
(%i31) s20[n]:=s01[n]+s03[n];
(%o31) s20n := s01n + s03n
(%i32) makelist(s20[n], n, 1, 5);
(%o32) [sin(1)+ $\frac{1}{5}$ , sin(2)+ $\frac{3}{8}$ , sin(3)+ $\frac{5}{11}$ , sin(4)+ $\frac{1}{2}$ , sin(5)+ $\frac{9}{17}$ ]
(%i33) s21[n]:=s01[n]-s03[n];
(%o33) s21n := s01n - s03n
(%i34) makelist(s21[n], n, 1, 5);
(%o34) [ $\frac{1}{5}$ -sin(1),  $\frac{3}{8}$ -sin(2),  $\frac{5}{11}$ -sin(3),  $\frac{1}{2}$ -sin(4),  $\frac{9}{17}$ -sin(5)]
(%i35) s22[n]:=2*s01[n]+3*n*s02[n];
(%o35) s22n := 2 s01n + 3 n s02n
(%i36) makelist(s22[n], n, 1, 5);
(%o36) [3 sin(1)+ $\frac{2}{5}$ , 6 sin( $\frac{1}{2}$ )+ $\frac{3}{4}$ , 9 sin( $\frac{1}{3}$ )+ $\frac{10}{11}$ , 12 sin( $\frac{1}{4}$ )+1, 15 sin( $\frac{1}{5}$ )+ $\frac{18}{17}$ ]
(%i37) s23[n]:=s02[n]*s03[n];
(%o37) s23n := s02n s03n
```


```
(%i38) makelist(s23[n], n, 1, 5);
(%o38) [ sin(1)^2, sin(1/2) sin(2), sin(1/3) sin(3), sin(1/4) sin(4), sin(1/5) sin(5)]

(%i39) s24[n]:=s05[n]/s06[n];
(%o39) s24_n := \frac{s05_n}{s06_n}

(%i40) makelist(s24[n], n, 1, 5);
(%o40) [-1, \frac{1}{8}, -\frac{1}{27}, \frac{1}{64}, -\frac{1}{125}]
```

Amb wxMaxima es pot obtenir una representació gràfica dels punts $(n, f(n))$ d'una successió, aspecte que pot ajudar a la seva interpretació, si més no, en una primera aproximació. La sintaxi és la que es pot veure a continuació i cal observar que s'han de definir dues llistes, una primera pels valors successius dels nombres naturals (abscisses) i una segona per als termes corresponents de la successió (ordenades):

```
(%i41) N[n]:=n$
List_0:makelist(N[n], n, 1, 40)$
(%i43) List_1:makelist(s01[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_1], [style, [points, 2,1,1]] , [xlabel, ""], [ylabel, ""])$
```




Si la instrucció que s'aplica és, com en aquest cas, “wxplot2d”, el gràfic s’enganxa a la sortida. En canvi, si la instrucció és “plot2d”, aleshores la sortida és un gràfic en un format anomenat “gnuplot”, que és un estàndard adoptat per wxMaxima; cal anar a la part superior del gràfic per modificar els paràmetres de la representació gràfica. Per copiar-lo i enganxar-lo després en un document, cal situar el cursor a la part superior del gràfic, clicar amb el botó dret del ratolí, anar a “options” i seleccionar “copy to clipboard”. Un cop fet això, cal anar al document i “enganxar” la imatge.

Un altre exemple. En aquest cas es tracta d'una successió de tipus “oscil·lant”:

```
(%i45) List_2:makelist(s03[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_2], [style, [points, 2,2,2]], [xlabel, ""], [ylabel, ""])$
```


(%t46)



Ara un exemple de successió divergent a $+\infty$:

```
(%i47) List_3:makelist(s06[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_3], [style,[points, 2,3,3]], [xlabel, ""], [ylabel, ""])$
```


(%t48)



Ara un exemple de successió divergent a $-\infty$:


```
(%i49) List_4:makelist(s07[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_4], [style, [points, 2,4,4]], [xlabel, ""], [ylabel, ""])$
```

(%t50)



I ara un exemple de successió convergent amb successions parcials de diferent comportament en la monotonia:

```
(%i56) List_5:makelist(s08[n], n, 1, 40)$
(%i51) List_5:makelist(s08[n], n, 1, 40)$
wxplot2d([discrete,List_0,List_5], [style, [points, 2,3,4]], [xlabel, ""], [ylabel, ""])$
```



En alguns casos, el terme general d'una successió conté un nombre de termes que depèn de l'índex n ; per exemple, la successió (a_n) definida per

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \cdots + \frac{1}{\sqrt{n^2+3n}}, \quad n \geq 1$$

Per tant, el terme a_n està definit amb la suma de n nombres; per tant, cada terme conté un nombre de termes que depèn de l'índex n . Amb wxMaxima aquesta successió es defineix mitjançant la sintaxi següent:

```
(%i53) a[n]:=sum(1/sqrt(n^2+3*k), k, 1, n);
a[1]; a[3]; a[5];
```

(%o53)
$$a_n := \sum_{k=1}^n \frac{1}{\sqrt{n^2+3k}}$$

(%o54)
$$\frac{1}{2}$$

(%o55)
$$\frac{1}{\sqrt{15}} + \frac{1}{2\sqrt{3}} + \frac{1}{3\sqrt{2}}$$

(%o56)
$$\frac{1}{\sqrt{37}} + \frac{1}{\sqrt{34}} + \frac{1}{\sqrt{31}} + \frac{1}{2\sqrt{10}} + \frac{1}{2\sqrt{7}}$$

Observeu que el programa escriu el terme n -èsim efectuant operacions de reordenació i racionalitzant algunes de les expressions que apareixen. Per exemple:

(%i57) a[10];

$$(\%o57) \frac{1}{\sqrt{130}} + \frac{1}{\sqrt{127}} + \frac{1}{\sqrt{118}} + \frac{1}{\sqrt{115}} + \frac{1}{\sqrt{109}} + \frac{1}{\sqrt{106}} + \frac{1}{\sqrt{103}} + \frac{1}{2\sqrt{31}} + \frac{1}{4\sqrt{7}} + \frac{1}{11}$$

(%i58) a[20];

$$(\%o58) \frac{1}{\sqrt{457}} + \frac{1}{\sqrt{454}} + \frac{1}{\sqrt{451}} + \frac{1}{\sqrt{445}} + \frac{1}{\sqrt{442}} + \frac{1}{\sqrt{439}} + \frac{1}{\sqrt{433}} + \frac{1}{\sqrt{430}} + \frac{1}{\sqrt{427}} + \frac{1}{\sqrt{421}} + \frac{1}{\sqrt{418}} + \frac{1}{\sqrt{415}} + \frac{1}{\sqrt{409}} + \frac{1}{\sqrt{406}} + \frac{1}{\sqrt{403}} + \frac{1}{2\sqrt{115}} + \frac{1}{2\sqrt{109}} + \frac{1}{2\sqrt{106}} + \frac{1}{2\sqrt{103}} + \frac{1}{8\sqrt{7}}$$

04-2.- Càlcul del límit d'una successió.

Els càlculs corresponents a aquest apartat es desenvolupen al fitxer **Tema_04-2.wxm**.

Començarem recordant la definició de límit d'una successió de nombres reals. Es diu que una successió $(x_n)_{n \in \mathbb{N}}$ té límit el nombre real $L \in \mathbb{R}$ si qualsevol que sigui $\varepsilon > 0$ existeix $n_0 \in \mathbb{N}$, que depèn de ε , tal que per a tot $n > n_0$ es compleix $|x_n - L| < \varepsilon$.

Una successió que té límit real es diu convergent; una successió que no té límit real es diu divergent. La notació habitual és $\lim(x_n) = L$ o bé $(x_n) \rightarrow L$. Les successions, en relació a si tenen o no límit, es classifiquen en:

- convergents, quan tenen límit real;
- divergents a infinit, quan el seu límit és $+\infty$ o $-\infty$.
- oscil·lants, quan no es compleix cap de les anteriors.

La interpretació de la definició de límit d'una successió és prou coneguda: tots els termes de la successió a partir del n_0 -èsim es troben en l'interval $]L - \varepsilon, L + \varepsilon[$.

Per comprovar que es compleix aquesta condició i il·lustrar-ho adequadament, ho farem amb un exemple. Considerem la successió de nombres reals $(x_n) = \left(\frac{2n-1}{3n+2}\right)_{n \in \mathbb{N}}$ i comprovem que és convergent i de límit $L = \frac{2}{3}$. En primer lloc calculem la diferència $|x_n - L|$:

```
(%i1) a[n]:=(2*n-1)/(3*n+2);  
(%o1) a_n :=  $\frac{2n-1}{3n+2}$   
(%i2) a[n]-2/3;  
(%o2)  $\frac{2n-1}{3n+2} - \frac{2}{3}$   
(%i3) ratsimp(a[n]-2/3);  
(%o3)  $-\frac{7}{9n+6}$   
(%i4) Difer1:-(ratsimp(a[n]-2/3));  
(%o4)  $\frac{7}{9n+6}$ 
```

Per tant, s'ha de complir l'equivalència següent:

$$\frac{7}{9n+6} < \varepsilon \quad \forall n > n_0 \quad \Leftrightarrow \quad \varepsilon - \frac{7}{9n+6} = \frac{9\varepsilon n + 6\varepsilon - 7}{9n+6} > 0 \quad \forall n > n_0$$

Aquesta condició equival a

$$9\epsilon n + 6\epsilon - 7 > 0 \quad \Leftrightarrow \quad n > \frac{7-6\epsilon}{9\epsilon}$$

Ara es tracta de definir les variables i aplicar el procediment amb wxMaxima; en podem veure uns exemples d'algunes iteracions:

```
(%i5) epsilon:0.01;          n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o5) 0.01
(%o6) 77.11111111111111
(%o7) [n > 77.11111111111111]
(%o8) true

(%i9) epsilon:0.0001;       n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o9) 1.0 10-4
(%o10) 7777.111111111111
(%o11) [n > 7777.111111111111]
(%o12) true

(%i13) epsilon:0.00000001;  n0:(7-6*epsilon)/(9*epsilon);
      assume(n>n0);          is(7/(9*n+6)<epsilon);
(%o13) 1.0 10-8
(%o14) 7.777777711111119 107
(%o15) [n > 7.777777711111119 107]
(%o16) true
```

Es pot anar disminuint el valor assignat a $\epsilon > 0$ i verificant que es compleix la condició a partir d'un nombre natural que es va incrementant a mesura que $\epsilon > 0$ va disminuint.

Quan es té definida una successió amb wxMaxima com una funció de variable discreta, una instrucció senzilla permet calcular el límit d'aquesta successió. Es tracta d'escriure:

limit(terme general de la successió, n, inf)

Vegem-ne alguns exemples.

Successió $\left(\frac{2n-1}{3n+2} \right)_{n \in \mathbb{N}}$:

```
(%i1) a[n]:=(2*n-1)/(3*n+2);
(%o1) a_n :=  $\frac{2n-1}{3n+2}$ 
(%i2) '(lim(a[n])) = limit(a[n], n, inf);
(%o2) lim(a_n) =  $\frac{2}{3}$ 
```

Successió $\left(\sin\left(\frac{1}{n}\right)\right)_{n \in \mathbb{N}}$:

```
(%i3) b[n]:=sin(1/n);
(%o3) b_n :=  $\sin\left(\frac{1}{n}\right)$ 
(%i4) '(lim(b[n])) = limit(b[n], n, inf);
(%o4) lim(b_n) = 0
```

Successió $(n^2 + 2n + 1)_{n \in \mathbb{N}}$:

```
(%i5) c[n]:=n^2+2*n+1;
(%o5) c_n :=  $n^2 + 2n + 1$ 
(%i6) '(lim(c[n])) = limit(c[n], n, inf);
(%o6) lim(c_n) =  $\infty$ 
```

Successió $(-n^2 + 2n + 1)_{n \in \mathbb{N}}$:

```
(%i7) d[n]:=-n^2+2*n+1;
(%o7) d_n :=  $-n^2 + 2n + 1$ 
(%i8) '(lim(d[n])) = limit(d[n], n, inf);
(%o8) lim(d_n) =  $-\infty$ 
```

Successió $((-1)^n)_{n \in \mathbb{N}}$:

```
(%i9) aa[n]:=(-1)^n;
(%o9) aa_n :=  $(-1)^n$ 
(%i10) '(lim(aa[n])) = limit(aa[n], n, inf);
(%o10) lim(aa_n) = ind
```

Observeu que en aquest cas el programa respon “ind” que significa que no es pot calcular el límit, ja que aquest no existeix.

Successió $\left(\left(1 + \frac{1}{n} \right)^n \right)_{n \in \mathbb{N}}$:

```
(%i11) e[n]:=(1+1/n)^n;
(%o11) e_n := \left( 1 + \frac{1}{n} \right)^n
(%i12) 'lim(e[n])) = limit(e[n], n, inf);
(%o12) lim(e_n) = %e
```

Val la pena recordar que es compleix la propietat següent: si $(x_n) \rightarrow +\infty$, aleshores es compleix:

$$\lim_{n \in \mathbb{N}} \left(\left(1 + \frac{1}{x_n} \right)^{x_n} \right) = e$$

A manera de comprovació:

```
(%i13) x[n]:=(1+1/n^2)^(n^2);
(%o13) x_n := \left( 1 + \frac{1}{n^2} \right)^{n^2}
(%i14) 'lim(x[n])) = limit(x[n], n, inf);
(%o14) lim(x_n) = %e
(%i15) y[n]:=(1+1/(3*n^2+5*n+4))^(3*n^2+5*n+4);
(%o15) y_n := \left( 1 + \frac{1}{3 n^2 + 5 n + 4} \right)^{3 n^2 + 5 n + 4}
(%i16) 'lim(y[n])) = limit(y[n], n, inf);
(%o16) lim(y_n) = %e
```

Límit d'una successió recurrent.

Per calcular el límit d'una successió definida per recurrència no es pot aplicar la sintaxi anterior. En efecte, considerem la successió recurrent definida per:

$$\begin{cases} x_1 = 3 \\ x_n = \sqrt{2 + x_{n-1}}, \quad \forall n \geq 2 \end{cases}$$

Si la posem en sintaxi de wxMaxima i calculem el límit, obtenim la resposta següent:

```
(%i1) x[1]:3;
      x[n]:=sqrt(2+x[n-1]);
(%o1) 3
      (%o2)  $x_n := \sqrt{2 + x_{n-1}}$ 
(%i3) '(lim(x[n])) = limit(x[n], n, inf);
Maxima encountered a Lisp error:
Error in PROGN [or a callee]: Bind stack overflow.
Automatically continuing.
To enable the Lisp debugger set *debugger-hook* to t
```

En aquest tipus de successions cal aplicar la metodologia següent: demostrar que és convergent aplicant els criteris de convergència, habitualment, monotonia i fitació. Vegeu l'apartat següent. Un cop s'ha demostrat que la successió és convergent, cal escriure l'equació que ha de complir el límit, definida a partir de la fórmula recurrent de la successió; en aquest cas l'equació és:

$$\lim(x_n) = \sqrt{2 + \lim(x_{n-1})} \quad \Leftrightarrow \quad L = \sqrt{2 + L}$$

En wxMaxima la sintaxi és la següent:

```
(%i4) Eq1:L=sqrt(2+L);
      solve(Eq1, L);
(%o4)  $L = \sqrt{L + 2}$ 
(%o5)  $[L = \sqrt{L + 2}]$ 
```

Com veiem, el programa no la pot resoldre en aquesta forma; cal escriure-la de la forma següent:

```
(%i6) Eq2:L^2=(2+L);
      solve(Eq2, L);
(%o6)  $L^2 = L + 2$ 
(%o7)  $[L = 2, L = -1]$ 
```

Aplicant les propietats de la successió, es pot afirmar que no té sentit que el límit sigui negatiu i, per tant, deduïm que $L = 2$.

Cal insistir en què aquesta metodologia tan sols és aplicable un cop s'ha demostrat que la successió és convergent. Per exemple, si es considera la successió recurrent definida per:

$$\begin{cases} x_1 = 1 \\ x_n = 2x_{n-1} + 1, \forall n \geq 2 \end{cases}$$

I es vol calcular el límit aplicant directament la metodologia descrita abans, es pot concloure que $\lim(x_n) = -1$:

```
(%i8) y[1]:1$
      y[n]:=2*y[n-1]+1;
(%o9) y_n:=2 y_{n-1} + 1
(%i10) Eq3:L=2*L+1;
      solve(Eq3, L);
(%o10) L=2 L+1
(%o11) [L=-1]
```

Aquesta conclusió és evidentment falsa, ja que és immediat comprovar que $x_n = 2^n - 1$, $n \geq 1$ i, per tant, que es compleix $\lim(x_n) = +\infty$.

En el cas de la successió de Fibonacci, definida per:

$$\begin{cases} x_1 = 1, & x_2 = 1 \\ x_n = x_{n-2} + x_{n-1}, & \forall n \geq 3 \end{cases}$$

és evident que es tracta d'una successió divergent a $+\infty$. Als exercicis es proposa una qüestió interessant en relació a aquesta successió (vegeu exercici 04.10).

04-3.- Formes indeterminades o indeterminacions

Els càlculs corresponents a aquest apartat es desenvolupen al fitxer **Tema_04-3.wxm**.

S'anomenen formes indeterminades o indeterminacions les successions construïdes amb operacions entre altres successions, de manera que inicialment no es pot afirmar res sobre el límit de la successió. Vegem uns exemples.

Exemple 04-3.1.- Considerem les successions numèriques:

$$x_n = (n^2 + 1); \quad y1_n = (n^2 - 1); \quad y2_n = (n^3 - 1); \quad y3_n = (n - 1)$$

Totes elles compleixen que tenen límit infinit. Calculem ara el límit de la successió diferència de la primera amb cada una de les altres:

```
(%i1) x[n]:=n^2+1; y1[n]:=n^2-1; y2[n]:=n^3-1; y3[n]:=n-1;
(%o1) x_n:=n^2+1
(%o2) y1_n:=n^2-1
(%o3) y2_n:=n^3-1
(%o4) y3_n:=n-1
(%i5) L1=limit(x[n]-y1[n], n, inf);
      L2=limit(x[n]-y2[n], n, inf);
      L3=limit(x[n]-y3[n], n, inf);
(%o5) L1=2
(%o6) L2=-∞
(%o7) L3=∞
```

Com es veu, en cada cas dona un resultat diferent, de manera que d'entrada no es pot afirmar res sobre quin serà el límit de la diferència de dues successions de límit infinit i, per tant, aquesta diferència és una forma indeterminada.

Exemple 04-3.2.- Considerem les successions numèriques:

$$x_n = \left(\frac{1}{n^2 + 1} \right); \quad y1_n = (n^2 - 1); \quad y2_n = (n^3 - 1); \quad y3_n = (n - 1)$$

La primera és un infinitèsim i les altres tres són infinits. Calculem ara el límit de la successió producte de la primera amb cada una de les altres:

```

(%i1) x[n]:=1/(n^2+1);  y1[n]:=n^2-1;
      y2[n]:=n^3-1;    y3[n]:=n-1;

(%o1)  $x_n := \frac{1}{n^2+1}$ 

(%o2)  $y1_n := n^2-1$ 

(%o3)  $y2_n := n^3-1$ 

(%o4)  $y3_n := n-1$ 

(%i5) L1=limit(x[n]*y1[n], n, inf);
      L2=limit(x[n]*y2[n], n, inf);
      L3=limit(x[n]*y3[n], n, inf);

(%o5) L1 = 1

(%o6) L2 =  $\infty$ 

(%o7) L3 = 0

```

Com es veu, en cada cas dóna un resultat diferent, de manera que d'entrada no es pot afirmar res sobre quin serà el límit del producte de dues successions essent la primera un infinitèsim i la segona un infinit i, per tant, aquest producte és una forma indeterminada. A la Taula 04-1 hi ha les formes indeterminades o indeterminacions per a successions numèriques i la metodologia de resolució, cas que no es pugui obtenir el límit amb wxMaxima.

Tipus d'indeterminació	Successions	Forma indeterminada	Metodologia de resolució
$\infty - \infty$	$(x_n) \rightarrow +\infty; (y_n) \rightarrow +\infty$	$\lim(x_n - y_n)$	Multiplicar i dividir per la suma $(x_n + y_n)$
$0 \cdot \infty$	$(a_n) \rightarrow 0; (x_n) \rightarrow +\infty$	$\lim(a_n x_n)$	Criteri de Stolz
$0/0$	$(a_n) \rightarrow 0; (b_n) \rightarrow 0$	$\lim(a_n / b_n)$	Criteri de Stolz
∞/∞	$(x_n) \rightarrow +\infty; (y_n) \rightarrow +\infty$	$\lim(x_n / y_n)$	Criteri de Stolz
1^∞	$(\lambda_n) \rightarrow 1; (x_n) \rightarrow +\infty$	$\lim(x_n^{\lambda_n})$	Reducció al nombre e
0^0	$(a_n) \rightarrow 0; (b_n) \rightarrow 0$	$\lim(a_n^{b_n})$	Aplicar logaritmes i després el criteri de Stolz
∞^0	$(a_n) \rightarrow 0; (x_n) \rightarrow +\infty$	$\lim(x_n^{a_n})$	Aplicar logaritmes i després el criteri de Stolz

Taula 04-1. Formes indeterminades per a successions numèriques

04-4.- Criteris de convergència i càlcul de límits de successions numèriques.

Els càlculs corresponents a aquest apartat es desenvolupen al fitxer **Tema_04-4.wxm**.

En aquest apartat es veuen alguns criteris de convergència i mètodes de càlcul de límits que s'apliquen més sovint en l'anàlisi de la convergència i en el càlcul de límits de successions.

Exemple 04-4.1 (Criteri de compressió). Es considera la successió $(a_n)_{n \in \mathbb{N}}$ de nombres reals definida per:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \dots + \frac{1}{\sqrt{n^2+3n}}, \quad n \geq 1$$

Si la introduïm en el programa wxMaxima i en volem calcular el límit obtenim:

```
(%i1) a[n]:=sum(1/sqrt(n^2+3*k), k, 1, n);
```

$$(\%o1) \ a_n := \sum_{k=1}^n \frac{1}{\sqrt{n^2+3k}}$$

```
(%i2) '(lim(a[n])) = limit(a[n], n, inf);
```

$$(\%o2) \ \lim(a_n) = \lim_{n \rightarrow \infty} \sum_{k=1}^n \frac{1}{\sqrt{n^2+3k}}$$

Per tant, no és possible fer el càlcul d'aquest límit de la manera senzilla habitual. Aleshores observem el següent: d'una banda es compleix:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \dots + \frac{1}{\sqrt{n^2+3n}} \leq \frac{1}{\sqrt{n^2+3}} + \dots + \frac{1}{\sqrt{n^2+3}} = \frac{n}{\sqrt{n^2+3}} = y_n$$

D'altra banda:

$$a_n = \frac{1}{\sqrt{n^2+3}} + \frac{1}{\sqrt{n^2+6}} + \dots + \frac{1}{\sqrt{n^2+3n}} \geq \frac{1}{\sqrt{n^2+3n}} + \dots + \frac{1}{\sqrt{n^2+3n}} = \frac{n}{\sqrt{n^2+3n}} = x_n$$

Ara podem calcular els límits d'aquestes successions amb wxMaxima:

```

(%i3) x[n]:=n/sqrt(n^2)$
(%i4) '(lim(x[n])) = limit(x[n], n, inf);
(%o4) lim(x_n)=1

(%i5) y[n]:=n/sqrt(n^2+3*n)$
(%i6) '(lim(y[n])) = limit(y[n], n, inf);
(%o6) lim(y_n)=1

```

Per tant, per aplicació del criteri de compressió, es compleix $\lim(a_n)=1$.

Exemple 04-4.2 (Teorema de la convergència monòtona). Es considera la successió $(a_n)_{n \in \mathbb{N}}$ de nombres reals definida per:

$$\begin{cases} a_1 = 3 \\ a_n = \sqrt{2 + a_{n-1}}, \quad n \geq 2 \end{cases}$$

Com abans, wxMaxima no dona el límit de la successió. En efecte:

```

(%i1) a[1]:3;      a[n]:=sqrt(2+a[n-1]);
(%o1) 3
(%o2) a_n:=sqrt(2+a_{n-1})
(%i3) makelist(a[n], n, 1, 5);
(%o3) [3, sqrt(5), sqrt(sqrt(5)+2), sqrt(sqrt(sqrt(5)+2)+2), sqrt(sqrt(sqrt(sqrt(5)+2)+2)+2)]
(%i4) '(lim(a[n])) = limit(a[n], n, inf);

```

Maxima encountered a Lisp error:

Error in PROGN [or a callee]: Bind stack overflow.

Automatically continuing.

To enable the Lisp debugger set *debugger-hook* to t

Calculem ara els primers termes d'aquesta successió:

```

(%i5) a[1]; a[2],numer; a[3],numer;
      a[4],numer; a[5],numer;
(%o5) 3
(%o6) 2.23606797749979
(%o7) 2.058171027271492
(%o8) 2.014490264873845
(%o9) 2.0036192914009

```

Això permet intuir que la successió és monòtona decreixent; en efecte, si se suposa que $a_n > a_{n-1}$ (hipòtesi d'inducció), aleshores es compleix $2 + a_n > 2 + a_{n-1}$ i per tant $\sqrt{2 + a_n} > \sqrt{2 + a_{n-1}}$, és a dir, $a_{n+1} > a_n$. Per tant, la successió $(a_n)_{n \in \mathbb{N}}$ és monòtona decreixent. Per veure que és fitada inferiorment, és suficient observar que per definició de la successió es compleix que $a_n > 0, \forall n \geq 1$. Per tant, la successió és convergent, en virtut del teorema de la convergència monòtona. Ara ja es pot aplicar la metodologia de càlcul del límit:

```
(%i10) Eq32:L=sqrt(2+L);
(%o10) L=sqrt(L+2)
(%i11) Eq32a:L^2=L+2;
      solve(Eq32a, L);
(%o11) L^2=L+2
(%o12) [L=2, L=-1]
```

En definitiva, es pot afirmar que $\lim(a_n) = 2$.

Exemple 04-4.3 (Criteri de l'arrel n -èsima). Es considera la successió $(a_n)_{n \in \mathbb{N}}$ de nombres reals definida per:

$$a_n = (2n^2 + 1)^{1/n}, n \geq 1$$

Definim la successió amb wxMaxima i calculem els primers termes d'aquesta successió:

```
(%i1) a[n]:=(2*n^2+1)^(1/n);
(%o1) a_n:=(2 n^2+1)^(1/n)
(%i2) [a[1], a[2],a[3],a[4],a[5]];
(%o2) [3, 3, 19^(1/3), 33^(1/4), 51^(1/5)]
(%i3) [a[1], a[2],a[3],a[4],a[5]], numer;
(%o3) [3, 3, 2.668401648721944, 2.39678172692843, 2.19540189742749]
```

Si construïm la successió $(b_n)_{n \in \mathbb{N}}$ amb $b_n = \frac{a_{n+1}}{a_n}$, es tindrà:

```
(%i4) b[n]:=a[n+1]/a[n];
(%o4) b_n:=a_n+1/a_n
```

```
(%i5) makelist(b[n], n, 1, 5);
(%o5) [1,  $\frac{19^{1/3}}{3}$ ,  $\frac{33^{1/4}}{19^{1/3}}$ ,  $\frac{51^{1/5}}{33^{1/4}}$ ,  $\frac{73^{1/6}}{51^{1/5}}$ ]
(%i6) makelist(float(b[n]), n, 1, 5);
(%o6) [1.0, 0.88946721624065, 0.89820875657021, 0.91597907008453, 0.93119315475983]
```

Calculem ara el límit d'aquesta successió:

```
(%i7) 'lim(b[n]) = limit(b[n], n, inf);
(%o7) lim(b_n) = 1
```

Per tant, es compleix $\lim(a_n) = 1$.

Exemple 04-4.4 (Criteri de Stolz). Es considera la successió $(a_n)_{n \in \mathbb{N}}$ de nombres reals definida per:

$$a_n = \frac{1 + \frac{1}{2} + \dots + \frac{1}{n}}{\log(n)}, \quad n \geq 2$$

Definim en primer lloc la successió del numerador:

```
(%i1) x[k]:=1/k$      X[n]:=sum(x[k], k, 1, n);
(%o2) X_n :=  $\sum_{k=1}^n x_k$ 
(%i3) makelist(X[n], n, 1, 10);
(%o3) [1,  $\frac{3}{2}$ ,  $\frac{11}{6}$ ,  $\frac{25}{12}$ ,  $\frac{137}{60}$ ,  $\frac{49}{20}$ ,  $\frac{363}{140}$ ,  $\frac{761}{280}$ ,  $\frac{7129}{2520}$ ,  $\frac{7381}{2520}$ ]
```

Ara la successió del denominador i la successió quocient:

```
(%i4) Y[n]:=log(n);
      a[n]:=X[n]/Y[n];
(%o4) Y_n := log(n)
(%o5) a_n :=  $\frac{X_n}{Y_n}$ 
```

Calculem el límit d'aquesta successió:

```
(%i6) '(lim(a[n])) = limit(a[n], n, inf);
```

$$\sum_{k=1}^n \frac{1}{k}$$

```
(%o6) lim(a_n) = lim_{n->inf} \frac{\sum_{k=1}^n \frac{1}{k}}{\log(n)}
```

Aquesta resposta del programa suggereix que no es pot calcular el límit de la manera estàndard. Aplicarem el criteri de Stolz, una vegada s'hagi verificat que es compleixen les condicions per a la seva aplicació:

```
(%i7) xx[n]:=x[n+1];      YY[n]:=Y[n+1]-Y[n];      aa[n]:=xx[n]/YY[n];
      makelist(aa[k], k, 1, 4);
```

```
(%o7) xx_n := x_{n+1}
```

```
(%o8) YY_n := Y_{n+1} - Y_n
```

```
(%o9) aa_n := \frac{xx_n}{YY_n}
```

```
(%o10) [\frac{1}{2 \log(2)}, \frac{1}{3 (\log(3) - \log(2))}, \frac{1}{4 (\log(4) - \log(3))}, \frac{1}{5 (\log(5) - \log(4))}]
```

Ara calculem el límit d'aquesta successió:

```
(%i11) '(lim(aa[n])) = limit(aa[n], n, inf);
```

```
(%o11) lim(aa_n) = 1
```

Per tant, es compleix $\lim(a_n) = 1$.

04-5.- Equivalència de successions

Els càlculs corresponents a aquest apartat es desenvolupen al fitxer **Tema_04-5.wxm**.

El concepte d'equivalència de successions i la seva aplicació al càlcul de límits es pot veure a les referències, particularment a [BU07] i [JB01]. En aquest apartat es mostren alguns exemples de successions equivalents i s'il·lustra l'aplicació d'aquest concepte al càlcul de límits de successions.

Recordem que dues successions numèriques $(a_n), (b_n)$ es diu que són equivalents, si, i només si, es compleix $\lim \frac{a_n}{b_n} = 1$. La notació habitual per designar aquest fet és

$(a_n) \sim (b_n)$. Per exemple, les successions numèriques $a_n = \frac{2n}{3n+1}$ i $b_n = \frac{2n+4}{3n+7}$ són equivalents. Si dues successions són convergents i equivalents, aleshores tenen el mateix límit; el recíproc no és cert, és a dir, dues successions poden ser convergents i del mateix límit però poden no ser equivalents, com per exemple les successions $a_n = \frac{2n}{3n^2+1}$ i $b_n = \frac{2n+4}{3n^4+7}$.

La propietat més interessant a efectes pràctics de les successions equivalents és la següent: si $(a_n) \sim (\alpha_n)$ i $L = \lim(a_n b_n)$ existeix, aleshores $L = \lim(\alpha_n b_n)$. És a dir, en productes i quocients es poden substituir alguns factors per factors equivalents.

Vegem a continuació alguns exemples de successions equivalents.

Exemple 04-5.1. Si (x_n) és un infinitèsim, és a dir és tal que $(x_n) \rightarrow 0$, aleshores es compleixen:

$$(x_n) \sim (\sin(x_n)) \quad ; \quad (x_n) \sim (\tan(x_n))$$

Vegem, per exemple, el cas en que $x_n = \frac{1}{n^2+1}$:

```
(%i1) x[n]:=1/(n^2+1)$      y[n]:=sin(x[n])$      a[n]:=x[n]/(y[n])$  
      'lim(a[n])=limit(a[n], n, inf);
```

$$(\%o4) \lim \left(\frac{1}{(n^2+1) \sin\left(\frac{1}{n^2+1}\right)} \right) = 1$$

```
(%i5) z[n]:=tan(x[n])$      b[n]:=x[n]/(z[n])$  
      'lim(b[n])=limit(b[n], n, inf);
```

$$(\%o7) \lim \left(\frac{1}{(n^2+1) \tan\left(\frac{1}{n^2+1}\right)} \right) = 1$$

Exemple 04-5.2. Si $(x_n) \rightarrow 0$ aleshores es compleix:

$$(x_n^2/2) \sim (1 - \cos(x_n))$$

Vegem-ne un exemple:

```
(%i1) x[n]:=1/(n^2+1)$ y[n]:=1-cos(x[n])$
      a[n]:=(y[n])/(x[n]^2/2);

(%o3) a_n := \frac{y_n}{\frac{x_n^2}{2}}

(%i4) 'lim(a[n])=limit(a[n], n, inf);

(%o4) \lim \left( 2 (n^2 + 1)^2 \left( 1 - \cos \left( \frac{1}{n^2 + 1} \right) \right) \right) = 1
```

Exemple 04-5.3. Si $(x_n) \rightarrow 0$ aleshores es compleixen:

$$(x_n) \sim (e^{x_n} - 1); \quad (1 + x_n) \sim (e^{x_n})$$

Vegem-ne un cas concret:

```
(%i1) x[n]:=1/(n^2+1)$ y[n]:=exp(x[n])-1$ a[n]:=(y[n])/x[n]$
      'lim(a[n])=limit(a[n], n, inf);

(%o4) \lim \left( (n^2 + 1) \left( e^{\frac{1}{n^2 + 1}} - 1 \right) \right) = 1

(%i5) xx[n]:=1+x[n]$ yy[n]:=exp(x[n])$ aa[n]:=yy[n]/xx[n]$
      'lim(aa[n])=limit(aa[n], n, inf);

(%o8) \lim \left( \frac{e^{\frac{1}{n^2 + 1}}}{\frac{1}{n^2 + 1} + 1} \right) = 1
```

Exemple 04-5.4. Si $(x_n) \rightarrow 0$ aleshores es compleix:

$$(x_n) \sim (\log(x_n + 1))$$

Vegem-ne un cas concret:

```
(%i1) x[n]:=1/(n^2+1)$      y[n]:=log(x[n]+1)$
(%i3) a[n]:=y[n]/x[n];
      'lim(a[n])=limit(a[n], n, inf);

(%o3) a_n := \frac{y_n}{x_n}

(%o4) \lim \left( (n^2 + 1) \log \left( \frac{1}{n^2 + 1} + 1 \right) \right) = 1
```

Exemple 04-5.5. Si $(y_n) \rightarrow 1$ aleshores es compleix:

$$(y_n - 1) \sim (\log(y_n))$$

Vegem-ne un cas concret:

```
(%i1) x[n]:=log(y[n]);      y[n]:=n^2/(n^2+1);
(%o1) x_n := log(y_n)

(%o2) y_n := \frac{n^2}{n^2 + 1}

(%i3) z[n]:=y[n]-1;      a[n]:=x[n]/z[n];
      'lim(a[n])=limit(a[n], n, inf);

(%o3) z_n := y_n - 1

(%o4) a_n := \frac{x_n}{z_n}

(%o5) \lim \left( \frac{\log \left( \frac{n^2}{n^2 + 1} \right)}{\frac{n^2}{n^2 + 1} - 1} \right) = 1
```

Exemple 04-5.6. Per calcular el límit

$$\lim \frac{3n \left(1 + \frac{1}{2} + \frac{1}{3} + \cdots + \frac{1}{n} \right)}{(n+2) \log \left(\frac{1}{n} \right) \cos \left(\frac{\pi n}{4n+1} \right)}$$

i implementant els càlculs amb wxMaxima, obtenim:

(%i1) x[n]:=(3*n)*sum(1/k, k, 1, n);

(%o1) $x_n := 3n \sum_{k=1}^n \frac{1}{k}$

(%i2) y[n]:=(n+2)*(log(1/n))*(cos((%pi*n)/(4*n+1)));

(%o2) $y_n := (n+2) \log\left(\frac{1}{n}\right) \cos\left(\frac{\pi n}{4n+1}\right)$

(%i3) a[n]:=x[n]/y[n]\$ '(lim(a[n])) = limit(a[n], n, inf);

(%o4) $\lim(a_n) = 3 \lim_{n \rightarrow \infty} \frac{\left(\sum_{k=1}^n \frac{1}{k} \right) n}{\log\left(\frac{1}{n}\right) (n+2) \cos\left(\frac{\pi n}{4n+1}\right)}$

Aplicant la metodologia indicada per a successions equivalents i tenint en compte l'Exemple 04-4.4, el límit anterior és el mateix que

$$\lim \frac{3n}{(n+2) \cos\left(\frac{\pi n}{4n+1}\right)}$$

Amb wxMaxima s'obté el resultat:

(%i5) b[n]:=(3*n)/((n+2)*(cos((%pi*n)/(4*n+1))));
limit(b[n], n, inf);

(%o5) $b_n := \frac{3n}{(n+2) \cos\left(\frac{\pi n}{4n+1}\right)}$

(%o6) $3\sqrt{2}$

Es poden aplicar les propietats de les successions equivalents en successions numèriques divergents a ∞ . Si la successió numèrica es polinòmica:

$$(a_p n^p + a_{p-1} n^{p-1} + \cdots + a_1 n + a_0)$$

l'equivalència més important i d'aplicació habitual és

$$(a_p n^p + a_{p-1} n^{p-1} + \cdots + a_1 n + a_0) \sim (a_p n^p)$$

és a dir, l'aplicació de la propietat:

$$\lim \frac{a_p n^p + a_{p-1} n^{p-1} + \dots + a_1 n + a_0}{a_p n^p} = 1$$

Finalment, una equivalència molt útil és la de Stirling: $(n!) \sim (n^n e^{-n} \sqrt{2\pi n})$. En efecte:

```
(%i1) a[n]:=n!;      b[n]:=(n^n)*(exp(-n))*sqrt(2*%pi*n);
(%o1) a_n:=n!
      (%o2) b_n:=n^n exp(-n) sqrt(2 pi n)
      (%i3) s[n]:=a[n]/b[n];
           '(lim(s[n])) = limit(s[n], n, inf);
(%o3) s_n:=a_n/b_n
      (%o4) lim(s_n)=1
```